

counter flow



a) Find convective heat transfer coefficient on hot & cold sides

$$\bar{Nu} = \frac{\bar{h} D_h}{k}$$

$$D_H = \frac{4A_c}{per}$$

Duct flow:



Aspect = H/W

→ KEY VAR IN EES

b) Fin resistance of hot & cold



→ KEY VARS IN EES

c) conductance (UA)

$$UA = \frac{1}{R_{tot}}$$



→ KEY VAR

d) NTU

$$NTU = \frac{UA}{\dot{C}_{min}}$$

→ KEY VAR

$$\dot{C}_{min} = \min\left(\dot{m}_C C_C, \dot{m}_H C_H\right)$$

e) effectiveness

$$\varepsilon = \frac{1 - \exp\left(-NTU(1 - C_R)\right)}{1 - C_R \exp\left(-NTU(1 - C_R)\right)}$$

$$C_R = \dot{C}_{min}/\dot{C}_{max}$$

→ ε KEY VAR

f)

$T_{C\,out}$, $T_{H\,out}$ KEY VARS

```
$unitsystem SI J Pa

"Geometry"
W = 5 [mm]*convert(mm,m)                    "Channel width"
th_f = 1 [mm]*convert(mm,m)                 "Fin width"
H = 10 [mm]*convert(mm,m)                   "Fin/Channel height"
th_b = 2 [mm]*convert(mm,m)                 "Channel separator thickness"

N_c = 500 [-]                               "Number of fins"

$ifnot parametrictable
L = 5 [m]                                   "Overall length of the heat exchanger"
$endif

T_H_in = 90 [C]                             "Hot stream inlet temperature"
T_C_in = 20 [C]                             "Cold stream inlet temperature"

m_dot_H_tot = 2 [kg/s]                      "Total hot side mass flow"
m_dot_C_tot = .5 [kg/s]                     "Total cold side mass flow"
m_dot_H = m_dot_H_tot / N_c                 "Single channel hot side mass flow"
m_dot_C = m_dot_C_tot / N_c                 "Single channel cold side mass flow"
T_prop = (T_H_in + T_C_in)/2                "Temperature used for property evaluations"

k_m=20 [W/m-K]                              "Metal conductivity"

"Property lookups for hot and cold side"
cp_H=cp(Air,T=T_prop)
cp_C =specheat(Water, T=T_prop, P=Po#)

rho_H = density(Air, T=T_prop, P=Po#)
rho_C = density(Water, T=T_prop, P=Po#)

mu_H = viscosity(Air, T=T_prop)
mu_C = viscosity(Water, T=T_prop, P=Po#)

k_C = conductivity(Water, T=T_prop, P=Po#)
k_H = conductivity(Air, T=T_prop)

"part a convective HTC"


"both"
D_h = 4*A_c/per
per = 2*H+2*W
LoverD_h = L/D_h
Aspect = W/H
RelRough = 0
A_c = W*H

"cold water"
Call ductflow_nd(Re_C,Pr_C,LoverD_h, Aspect,RelRough: Nusselt_T_C, Nusselt_H_C, f_C)
u_C = m_dot_C/rho_C /A_c
Re_C = rho_C*u_C*D_h/mu_C
Pr_C = mu_C*cp_C/k_C
Nusselt_H_C = h_bar_C*D_h/k_C

"hot air"
```

```
Call ductflow_nd(Re_H,Pr_H,LoverD_h, Aspect,RelRough: Nusselt_T_H, Nusselt_H_H, f_H)
u_H = m_dot_H/rho_H /A_c
Re_H = rho_H*u_H*D_h/mu_H
Pr_H = mu_H*cp_H/k_H
Nusselt_H_H = h_bar_H*D_h/k_H


"part b fin resistance"


//per_f = 2*H+th_f
per_f = 2*L                                   "TA says perimeter from top down"
A_f = th_f*L                                  "top down area"
A_s = 2*H*L

"cold fin"
m_C = sqrt(per_f*h_bar_C/(k_m*A_f))
eta_fin_C = tanh(m_C*H)/(m_C*H)
finres_C = 1/(eta_fin_C*h_bar_C*A_s)

"hot fin"
m_H = sqrt(per_f*h_bar_H/(k_m*A_f))
eta_fin_H = tanh(m_H*H)/(m_H*H)
finres_H = 1/(eta_fin_H*h_bar_H*A_s)


"part c conductance"


A_unf = W*L
R_conv_H = 1/(h_bar_H*A_unf)
R_conv_C = 1/(h_bar_C*A_unf)
R_cond = th_b/(k_m*A_unf)

R_tot = R_cond/N_c + 1/( N_c/R_conv_C+N_c/finres_C ) + 1/( N_c/R_conv_H+N_c/finres_H )
UA = 1/R_tot


"part d NTU"


c_dot_C = m_dot_C_tot*cp_C
c_dot_H = m_dot_H_tot*cp_H
c_dot_min = min(c_dot_C,c_dot_H)

NTU = UA/c_dot_min


"part e effectiveness"


C_R = min(c_dot_C,c_dot_H)/max(c_dot_C,c_dot_H)
epsilon = (1-exp(-NTU*(1-C_R))) / (1-C_R*exp(-NTU*(1-C_R)))


"part f outlet temps"
```

T_C_out = T_C_in + epsilon*c_dot_min*(T_H_in-T_C_in)/c_dot_C
T_H_out = T_H_in - epsilon*c_dot_min*(T_H_in-T_C_in)/c_dot_H

SOLUTION

Unit Settings: SI C Pa J mass deg

$Aspect = 0.5$ [-]	$A_c = 0.00005$ [m^2]	$A_f = 0.005$ [m^2]	$A_s = 0.1$ [m^2]
$A_{unf} = 0.025$ [m^2]	$cp_C = 4183$ [J/kg-K]	$cp_H = 1006$ [J/kg-K]	$\dot{c}_C = 2091$ [J/K-s]
$\dot{c}_H = 2013$ [J/K-s]	$\dot{c}_{min} = 2013$ [W/C]	$C_R = 0.9624$ [-]	$D_h = 0.006667$ [m]
$\varepsilon = 0.7634$ [-]	$\eta_{fin,C} = 0.4679$ [-]	$\eta_{fin,H} = 0.5623$	$finres_C = 0.04986$ [C/W]
$finres_H = 0.06512$ [C/W]	$f_C = 0.2367$ [-]	$f_H = 0.02434$ [-]	$H = 0.01$ [m]
$\bar{h}_C = 428.7$ [W/m^2-K]	$\bar{h}_H = 273.1$ [W/m^2-K]	$k_C = 0.646$ [W/m-K]	$k_H = 0.02772$ [W/m-K]
$k_m = 20$ [W/m-K]	$L = 5$ [m]	$LoverD_h = 750$ [-]	$\mu_C = 0.0005036$ [Pa-s]
$\mu_H = 0.00001986$ [Pa-s]	$m_C = 207$ [1/m]	$\dot{m}_C = 0.001$ [kg/s]	$\dot{m}_{C,tot} = 0.5$ [kg/s]
$\dot{m}_H = 0.004$ [kg/s]	$\dot{m}_{H,tot} = 2$ [kg/s]	$m_H = 165.3$ [1/m]	$NTU = 3.045$ [-]
$Nusselt_{H,C} = 4.424$ [-]	$Nusselt_{H,H} = 65.68$ [-]	$Nusselt_{T,C} = 3.539$ [-]	$Nusselt_{T,H} = 65.68$ [-]
$N_c = 500$ [-]	$per = 0.03$ [m]	$per_f = 10$ [m]	$Pr_C = 3.261$ [-]
$Pr_H = 0.721$ [-]	$RelRough = 0$ [-]	$Re_C = 264.7$ [-]	$Re_H = 26856$ [-]
$\rho_C = 985.7$ [kg/m^3]	$\rho_H = 1.076$ [kg/m^3]	$R_{cond} = 0.004$ [C/W]	$R_{conv,C} = 0.09331$ [C/W]
$R_{conv,H} = 0.1465$ [C/W]	$R_{tot} = 0.0001632$ [C/W]	$th_b = 0.002$ [m]	$th_f = 0.001$ [m]
$T_{C,in} = 20$ [C]	$T_{C,out} = 71.43$ [C]	$T_{H,in} = 90$ [C]	$T_{H,out} = 36.56$ [C]
$T_{prop} = 55$ [C]	$UA = 6129$ [W/C]	$u_C = 0.02029$ [m/s]	$u_H = 74.37$ [m/s]
$W = 0.005$ [m]			

No unit problems were detected.

KEY VARIABLES

$\bar{h}_H = 273.1$ [W/m^2-K]	*a) HTC for hot air side*
$\bar{h}_C = 428.7$ [W/m^2-K]	*a) HTC for cold water side*
$finres_C = 0.04986$ [C/W]	*b) fin resistance for cold*
$finres_H = 0.06512$ [C/W]	*b) fin resistance for hot*
$UA = 6129$ [W/C]	*c) total conductance*
$NTU = 3.045$ [-]	*d) NTU*
$\varepsilon = 0.7634$ [-]	*e) heat exchanger effectiveness*
$T_{C,out} = 71.43$ [C]	*f) cold outlet temp*
$T_{H,out} = 36.56$ [C]	*f) hot outlet temp*