

a) Calculate R^g_{12} & R^g_{13}

$\rightarrow R^g_{12} > R^g_{13}$, expect surface 3 to receive more (less resistance)

b) Surf. Res. $\rightarrow$ EES

c) Res. network for 2



d) Sys of eq. to solve for T & q, Find $\dot{q}_3$

$$\dot{q}_i = \frac{E_{b_i} - J_i}{R^s_i} \qquad E_{b_i} = \sigma T_i^4$$

$$J_i = (1-\varepsilon_i) G_i + \varepsilon_i E_{b_i}$$

$\rightarrow$ EES
$$\dot{q}_{ij} = \frac{(J_i - J_j)}{R^g_{ij}}$$

$$R^g_{i,j} = \frac{1}{A_i F_{ij}}$$

e) plot $\rightarrow$ EES

```
$unitsystem SI K J Pa
$tabstops 0.2 0.4 0.6 2

"Given information"
I_e = 2.5 [A]                      "Circuit electricity draw"
V_e = 120 [V]                      "Voltage drop"
k = 0.5 [W/m-K]                    "Base layer thermal conductivity"

epsilon[1] = 0.9 [-]               "Resistor surface emissivity"
epsilon[2] = 0.3 [-]               "Bottom surface emissivity"
epsilon[3] = 0.6 [-]               "Bread emissivity"
epsilon[4] = 1 [-]                 "Surroundings emissivity"

"Temperatures"
$varinfo T[] units='K'
$ifnot parametrictable
T[3] = converttemp(C, K, 25 [C])
$end
T[4] = converttemp(C, K, 18 [C])
T_b = converttemp(C, K, 18 [C])

"Geometry"
a = 1.5 [cm]*convert(cm,m)
H = 12 [cm]*convert(cm,m)
L = 15 [cm]*convert(cm,m)
s = 2 [cm]*convert(cm,m)

"Areas"
$varinfo A[] units='m^2'
A[1] = H*L
A[2] = s*L
A[3] = H*L
A[4] = H*L*100000

"View factors between all surfaces (assume a 2-D problem)"
$varinfo F[] units='-'
F[1,1] = 0
F[1,2] = F[2,1]*A[2]/A[1]
F[1,3] = f2d_1(s,H)
F[1,4] = 1 - (F[1,2] + F[1,3])

F[2,1] = f2d_12(s,H)
F[2,2] = 0
F[2,3] = F[2,1]
F[2,4] = 1 - sum(F[2,j],j=1,3)

F[3,1] = F[1,3]
F[3,2] = F[1,2]
F[3,3] = 0
F[3,4] = F[1,4]

Duplicate i=1,4
    F[4,i] = F[i,4]*A[i]/A[4]
End

"-----------------------------------------------------------------------"
```

```
"a) geometric resistances, same in both directions"
R|g[1,2] = 1/(A[1]*F[1,2])
R|g[2,1] = R|g[1,2]
R|g[1,3] = 1/(A[1]*F[1,3])
R|g[3,1] = R|g[1,3]
R|g[1,4] = 1/(A[1]*F[1,4])
R|g[4,1] = R|g[1,4]
R|g[2,3] = 1/(A[2]*F[2,3])
R|g[3,2] = R|g[2,3]
R|g[2,4] = 1/(A[2]*F[2,4])
R|g[4,2] = R|g[2,4]
R|g[3,4] = 1/(A[3]*F[3,4])
R|g[4,3] = R|g[3,4]

"b) surface resistance"
$varinfo R|s[] units='1/m^2'
$varinfo R|g[] units='1/m^2'

Duplicate i=1,4
    R|s[i]=(1-epsilon[i])/(A[i]*epsilon[i])
End

"d) solve for t and q"
R_cond = a/(k*s*L)
q_dot[1] = I_e*V_e
q_dot[2] = (T_b-T[2])/R_cond

q_dot[1]*R|s[1] = ( E_b[1] - J[1] )
q_dot[2]*R|s[2] = ( E_b[2] - J[2] )
q_dot[3]*R|s[3] = ( E_b[3] - J[3] )
q_dot[4]*R|s[4] = ( E_b[4] - J[4] )

$varinfo E_b[] units='W/m^2'
E_b[1] = sigma#*T[1]^4
E_b[2] = sigma#*T[2]^4
E_b[3] = sigma#*T[3]^4
E_b[4] = sigma#*T[4]^4

//J[1] = (1-epsilon[1])*G[1]+epsilon[1]*E_b[1]
//J[2] = (1-epsilon[2])*G[2]+epsilon[2]*E_b[2]
//J[3] = (1-epsilon[3])*G[3]+epsilon[3]*E_b[3]
//J[4] = (1-epsilon[4])*G[4]+epsilon[4]*E_b[4]

q_dot[1,2]*R|g[1,2] = (J[1]-J[2])
q_dot[1,3]*R|g[1,3] = (J[1]-J[3])
q_dot[1,4]*R|g[1,4] = (J[1]-J[4])
q_dot[1] = q_dot[1,2]+q_dot[1,3]+q_dot[1,4]

q_dot[2,1]*R|g[2,1] = (J[2]-J[1])
q_dot[2,3]*R|g[2,3] = (J[2]-J[3])
q_dot[2,4]*R|g[2,4] = (J[2]-J[4])
q_dot[2] = q_dot[2,1]+q_dot[2,3]+q_dot[2,4]

q_dot[3,1]*R|g[3,1] = (J[3]-J[1])
q_dot[3,2]*R|g[3,2] = (J[3]-J[2])
q_dot[3,4]*R|g[3,4] = (J[3]-J[4])
```

```
q_dot[3] = q_dot[3,1]+q_dot[3,2]+q_dot[3,4]

q_dot[4,1]*R|g[4,1] = (J[4]-J[1])
q_dot[4,2]*R|g[4,2] = (J[4]-J[2])
q_dot[4,3]*R|g[4,3] = (J[4]-J[3])
q_dot[4] = q_dot[4,1]+q_dot[4,2]+q_dot[4,3]

"e) plot of efficiency with varying ambient temp"
eta = -q_dot[3]/q_dot[1]
T0 = T[4]
Tf = converttemp(C,K,155[C])
```

SOLUTION

Unit Settings: SI K Pa J mass deg

a = 0.015 [m]	η = 0.7927 [-]	H = 0.12 [m]	I_e = 2.5 [A]	k = 0.5 [W/m-K]
L = 0.15 [m]	R_{cond} = 10 [K/W]	s = 0.02 [m]	T0 = 291.2 [K]	Tf = 428.2 [K]
T_b = 291.2 [K]	V_e = 120 [V]			

No unit problems were detected.

KEY VARIABLES

R_1^s = 6.173 [1/m^2]	*b) surf res*
R_2^s = 777.8 [1/m^2]	*b) surf res*
R_3^s = 37.04 [1/m^2]	*b) surf res*
R_4^s = 0 [1/m^2]	*b) surf res*
$\dot{q}_3$ = -237.8 [W]	*d) q_dot_3*


0.795
0.79
0.785
0.78
0.775
η [-]
250
300
350
400
450
T[3] [K]